Simpson's Rule and Cubics

Mth 351 July 14 1999
Bent E. Petersen
Filename: simpson_cubic.mws

Degree 3 and 2

It is well known that Simpson's quadrature is more precise than one at first expects and moreover is exact for cubics. The second fact accounts for the first one. We can use Maple to demonstrate symbolically that Simpson's rule is exact for cubics. Morever we can demonstrate a similar phenomenon for higher degree polynomials and other compound quadrature rules based on interpolation.

To show that Simpson's rule is exact for cubics it suffices to show that for any three points on a cubic, with equispaced abscissas, the area under the graph of the cubic from the first to the last point is the same as the area under the graph, on the same interval, of the unique quadratic throught the three points.

We begin by constructing a general cubic polynomial (function)

```
> p3:=unapply(a*x^3+b*x^2+c*x+d,x);
```

$$p3 := x \to a\,x^3 + b\,x^2 + c\,x + d$$

Now let's compute the integral from $r - h$ to $r + h$ for any real numbers r and h.

```
> A3:=collect(int(p3(x),x=r-h..r+h),h);
```

$$A3 := \left(\frac{2}{3}b + 2\,a\,r\right)h^3 + (2\,a\,r^3 + 2\,d + 2\,b\,r^2 + 2\,c\,r)\,h$$

Now we compute the quadratic interpolation polynomial *p2* through the points on the graph of *p3* with abscissas $r - h$, r and $r + h$.

```
> X2:=[r-h,r,r+h];
```

$$X2 := [r - h, r, r + h]$$

```
> Y2:=map(p3,X2);
```

$$Y2 := [a\,(r-h)^3 + b\,(r-h)^2 + c\,(r-h) + d,\ a\,r^3 + b\,r^2 + c\,r + d,\ a\,(r+h)^3 + b\,(r+h)^2 + c\,(r+h) + d]$$

```
> q2:=unapply(interp(X2,Y2,x),x);
```

$$q2 := x \to (3\,a\,r + b)\,x^2 + (a\,h^2 - 3\,a\,r^2 + c)\,x + a\,r^3 - a\,r\,h^2 + d$$

Let's compute the integral from $r - h$ to $r + h$ for any real numbers r and h.

```
> B2:=collect(int(q2(x),x=r-h..r+h),h);
```

$$B2 := \left(\frac{2}{3}b + 2\,a\,r\right)h^3 + (2\,(3\,a\,r + b)\,r^2 + 2\,(-3\,a\,r^2 + c)\,r + 2\,a\,r^3 + 2\,d)\,h$$

```
> simplify(A3-B2);
```

$$0$$

It is now easy to see that Simpson's compound quadrature rule, that is, Simpson's rule, is exact for cubics.

Degree 5 and 4

Let's verify that similar results hold for higher degree. We first show that given any quintic polynomial (degree 5) and five points on its graph with equispaced abscissas, the area under the graph of the quintic from the first to the last point is the same as the area under the graph, over the same interval, of the unique quartic (degree 4) interpolation polynomial through the given points.

We begin by constructing a general quintic polynomial.

```
> p5:=unapply(a*x^5+b*x^4+c*x^3+d*x^2+e*x+f,x);
```

$$p5 := x \to a\,x^5 + b\,x^4 + c\,x^3 + d\,x^2 + e\,x + f$$

Now let's compute the integral from $r - 2\,h$ to $r + 2\,h$ for any real numbers r and h.

```
> A5:=collect(int(p5(x),x=r-2*h..r+2*h),h);
```

$$A5 := \left(\frac{64}{5}\,b + 64\,a\,r\right)h^5 + \left(\frac{160}{3}\,a\,r^3 + \frac{16}{3}\,d + 32\,b\,r^2 + 16\,c\,r\right)h^3 + (4\,a\,r^5 + 4\,c\,r^3 + 4\,b\,r^4 + 4\,d\,r^2 + 4\,e\,r + 4\,f)\,h$$

Now we compute the quartic interpolation polynomial *p4* through the points on the graph of *p5* with abscissas $r - 2\,h$, $r - h$, r, $r + h$ and $r + 2\,h$.

```
> X4:=[r-2*h,r-h,r,r+h,r+2*h];
```

$$X4 := [r - 2\,h, r - h, r, r + h, r + 2\,h]$$

```
> Y4:=map(p5,X4);
```

$$Y4 := [a\,(r - 2\,h)^5 + b\,(r - 2\,h)^4 + c\,(r - 2\,h)^3 + d\,(r - 2\,h)^2 + e\,(r - 2\,h) + f,\\ a\,(r - h)^5 + b\,(r - h)^4 + c\,(r - h)^3 + d\,(r - h)^2 + e\,(r - h) + f, a\,r^5 + b\,r^4 + c\,r^3 + d\,r^2 + e\,r + f,\\ a\,(r + h)^5 + b\,(r + h)^4 + c\,(r + h)^3 + d\,(r + h)^2 + e\,(r + h) + f,\\ a\,(r + 2\,h)^5 + b\,(r + 2\,h)^4 + c\,(r + 2\,h)^3 + d\,(r + 2\,h)^2 + e\,(r + 2\,h) + f]$$

```
> q4:=unapply(interp(X4,Y4,x),x);
```

$$q4 := x \to (b + 5\,a\,r)\,x^4 + (c - 10\,a\,r^2 + 5\,a\,h^2)\,x^3 + (-15\,a\,r\,h^2 + 10\,a\,r^3 + d)\,x^2 + (-4\,a\,h^4 + e + 15\,a\,r^2\,h^2 - 5\,a\,r^4)\,x - 5\,a\,r^3\,h^2 + f + a\,r^5 + 4\,a\,r\,h^4$$

Let's compute the integral from $r - 2\,h$ to $r + 2\,h$ for any real numbers r and h.

```
> B4:=collect(int(q4(x),x=r-2*h..r+2*h),h);
```

$$B4 := \left(\frac{64}{5}\,b + 64\,a\,r\right)h^5 + \left(16\,(c - 10\,a\,r^2)\,r + \frac{160}{3}\,a\,r^3 + \frac{16}{3}\,d + 32\,(b + 5\,a\,r)\,r^2\right)h^3 + (4\,(b + 5\,a\,r)\,r^4 + 4\,a\,r^5 + 4\,(10\,a\,r^3 + d)\,r^2 + 4\,(c - 10\,a\,r^2)\,r^3 + 4\,(e - 5\,a\,r^4)\,r + 4\,f)\,h$$

```
> simplify(A5-B4);
```

$$0$$

The algebra would be tedious (and error prone) to do by hand, but Maple has no problem with it, so let's try one more example.

Degree 7 and 6

Now let's verify that similar results hold for degree 7 and 6. We show that given any polynomial of degree 7 and seven points on its graph with equispaced abscissas, the area under the graph of the degree 7 polynomial from the first to the last point is the same as the area under the graph, over the same interval, of the unique interpolation polynomial of degree 6 through the given points.

We begin by constructing the general polynomial of degree 7.

```
>
> p7:=unapply(a*x^7+b*x^6+c*x^5+d*x^4+e*x^3+f*x^2+g*x^1+h,x);
```

$$p7 := x \to a\,x^7 + b\,x^6 + c\,x^5 + d\,x^4 + e\,x^3 + f\,x^2 + g\,x + h$$

Now let's compute the integral from $r - 3\,h$ to $r + 3\,h$ for any real numbers r and h.

```
> A7:=collect(int(p7(x),x=r-3*h..r+3*h),h);
```

$$A7 := \left(4374\,a\,r + \frac{4374}{7}\,b\right)h^7 + \left(3402\,a\,r^3 + 1458\,b\,r^2 + \frac{486}{5}\,d + 486\,c\,r\right)h^5 + (54\,e\,r + 180\,c\,r^3 + 378\,a\,r^5 + 108\,d\,r^2 + 18\,f + 270\,b\,r^4)\,h^3 + 6\,h^2 + (6\,a\,r^7 + 6\,e\,r^3 + 6\,f\,r^2 + 6\,b\,r^6 + 6\,g\,r + 6\,d\,r^4 + 6\,c\,r^5)\,h$$

Now we compute the interpolation polynomial *p6* of degree 6 through the points on the graph of *p7* with abscissas $r - 3\,h$, $r - 2\,h$, $r - h$, r, $r + h$, $r + 2\,h$ and $r + 3\,h$.

```
> X6:=[r-3*h,r-2*h,r-h,r,r+h,r+2*h,r+3*h];
```

$$X6 := [r - 3\,h, r - 2\,h, r - h, r, r + h, r + 2\,h, r + 3\,h]$$

```
> Y6:=map(p7,X6);
```

$$Y6 := [a\,(r - 3\,h)^7 + b\,(r - 3\,h)^6 + c\,(r - 3\,h)^5 + d\,(r - 3\,h)^4 + e\,(r - 3\,h)^3 + f\,(r - 3\,h)^2 + g\,(r - 3\,h) + h,\; a\,(r - 2\,h)^7 + b\,(r - 2\,h)^6 + c\,(r - 2\,h)^5 + d\,(r - 2\,h)^4 + e\,(r - 2\,h)^3 + f\,(r - 2\,h)^2 + g\,(r - 2\,h) + h,$$
$$a\,(r - h)^7 + b\,(r - h)^6 + c\,(r - h)^5 + d\,(r - h)^4 + e\,(r - h)^3 + f\,(r - h)^2 + g\,(r - h) + h,$$
$$a\,r^7 + b\,r^6 + c\,r^5 + d\,r^4 + e\,r^3 + f\,r^2 + g\,r + h,$$
$$a\,(r + h)^7 + b\,(r + h)^6 + c\,(r + h)^5 + d\,(r + h)^4 + e\,(r + h)^3 + f\,(r + h)^2 + g\,(r + h) + h,$$
$$a\,(r + 2\,h)^7 + b\,(r + 2\,h)^6 + c\,(r + 2\,h)^5 + d\,(r + 2\,h)^4 + e\,(r + 2\,h)^3 + f\,(r + 2\,h)^2 + g\,(r + 2\,h) + h,\; a\,(r + 3\,h)^7 + b\,(r + 3\,h)^6 + c\,(r + 3\,h)^5 + d\,(r + 3\,h)^4 + e\,(r + 3\,h)^3 + f\,(r + 3\,h)^2 + g\,(r + 3\,h) + h]$$

```
> q6:=unapply(interp(X6,Y6,x),x);
```

$$q6 := x \to (7\,a\,r + b)\,x^6 + (-21\,a\,r^2 + c + 14\,a\,h^2)\,x^5 + (d - 70\,a\,r\,h^2 + 35\,a\,r^3)\,x^4 + (-35\,a\,r^4 + e - 49\,a\,h^4 + 140\,a\,r^2\,h^2)\,x^3 + (f + 21\,a\,r^5 - 140\,a\,r^3\,h^2 + 147\,a\,r\,h^4)\,x^2 + (36\,a\,h^6 + g + 70\,a\,r^4\,h^2 - 147\,a\,r^2\,h^4 - 7\,a\,r^6)\,x + h + 49\,a\,r^3\,h^4 - 36\,a\,r\,h^6 + a\,r^7 - 14\,a\,r^5\,h^2$$

Let's compute the integral from $r - 3\,h$ to $r + 3\,h$.

```
> B6:=collect(int(q6(x),x=r-3*h..r+3*h),h);
```

$$B6 := \left(4374\,a\,r + \frac{4374}{7}\,b\right)h^7$$

$$+\left(486\,(-21\,a\,r^2+c)\,r+3402\,a\,r^3+\frac{486}{5}\,d+1458\,(7\,a\,r+b)\,r^2\right)h^5+(378\,a\,r^5$$
$$+180\,(-21\,a\,r^2+c)\,r^3+270\,(7\,a\,r+b)\,r^4+108\,(d+35\,a\,r^3)\,r^2+54\,(-35\,a\,r^4+e)\,r+18\,f)\,h^3$$
$$+6\,h^2+(6\,(7\,a\,r+b)\,r^6+6\,a\,r^7+6\,(f+21\,a\,r^5)\,r^2+6\,(-21\,a\,r^2+c)\,r^5+6\,(-7\,a\,r^6+g)\,r$$
$$+6\,(d+35\,a\,r^3)\,r^4+6\,(-35\,a\,r^4+e)\,r^3)\,h$$

```
> simplify(A7-B6);
```

$$0$$

Are you beginning to suspect a general fact? Note the direct approach taken above is not suitable for hand calculation even for fairly low degree. Eventually Maple will give up too. But no finite number of cases will suffice to prove the general fact anyway. For a proof, what we really need is a suitable error estimate for the compound interpolation quadrature rules with equispaced nodes.